Appendix S1: Example model guide

Introduction

FcεRI is the high-affinity receptor for IgE antibody. On rat basophilic leukemia (RBL) cells, FcεRI is tetrameric. The α subunit binds IgE, whereas the β subunit and a homodimer of disulfide-linked γ subunits initiate signaling inside the cell. The signaling subunits contain immunoreceptor tyrosine-based activation motifs (ITAMs), which are phosphorylated by Src-family kinases and then serve as binding sites for SH2 domain-containing proteins. The model of Goldstein et al. [1] and Faeder et al. [2], which is annotated here, encompasses early events in FcεRI signaling in RBL cells: ligand-receptor binding, receptor aggregation, phosphorylation of ITAMs, recruitment of the kinases Lyn and Syk, and phosphorylation of Syk. The model contains extracellular (`E`), plasma membrane (`M`), and cytoplasmic (`C`) compartments. An extended contact map for this model (Fig. 4 in the main text) is shown below.



Figure 4: Extended contact map

Lines with arrowheads on both ends indicate non-covalent binding, open circles point to substrates of enzymes, and small squares indicate the positions of phosphorylated tyrosine residues within proteins. Compartments are indicated in tabs below the lower left corners of molecule boxes: E = extracellular, M = membrane, C = cytoplasmic. Only protein components included in the model are shown. See the 'Notes on Proteins Included in Model' section of this appendix for more complete information about protein substructures.

Table S1: Molecules of interest

Protein	UniProt ID	Gene	Comments
IgE	-	-	Covalently crosslinked antibody homodimer
FcεRI, α subunit	P12371	Fcer1a	Type I transmembrane protein
FcεRI, β subunit	P13386	Fcer1b	Type II transmembrane protein
FcεRI, γ subunit	P20411	Fcer1g	Type I transmembrane protein
Lyn	Q07014	Lyn	Src-family protein tyrosine kinase
Syk	Q64725	Syk	Syk-family protein tyrosine kinase

Notes on Proteins Included in Model

1. **IgE (Immunoglobulin E).** A chemically crosslinked dimer of IgE is a ligand that functions as a signal for mast cell degranulation [3–5]. Each IgE antibody contains an Fc region, which interacts with the α subunit of a receptor to initiate signaling.

   In the model (model.bngl, ESI) IgE (`Lig`) is assigned the following molecule type definition:

   ```
   Lig(Fc,Fc)
   ```
   (1)

   The component `Fc` represents the Fc region of an antibody. The two IgE antibodies in the dimer are identical; accordingly, the molecule `Lig` contains two identical `Fc` components. In the absence of interaction with other molecules, `Lig` is located in the extracellular compartment, `E` (Table S2). The total abundance of `Lig` is `Lig_tot` (Table S3).

   In the extended contact map (Fig. 4), IgE is represented by a simplified form of the following annotated hierarchical molecule type graph:



2. **FcεRI (high-affinity receptor for IgE)**

   FcεRI is a member of the family of multichain immune recognition receptors (MIRRs). The tetrameric form, which is considered in this model, contains an α chain, a β chain, and a homodimer of γ chains. The α chain interacts with IgE. The β and γ chains initiate signaling within the cell; these subunits contain immunoreceptor tyrosine-based activation motifs (ITAMs) that are phosphorylated as part of the earliest events in signaling.

   (a) α **chain.** The α chain is a member of the immunoglobulin superfamily. It contains two extracellular immunoglobulin-related domains that bind the Fc regions of an IgE antibody. The α chain also contains a transmembrane domain and a short cytoplasmic region [6]:



   In the above diagram, numbering of amino acid residues is based on UniProt entry P12371 (uniprot.org).

   (b) β **chain.** Experimental evidence indicates that the β chain functions as an amplifier of signaling [7]. This subunit contains an unusual ITAM, being shorter than the canonical sequence and having an additional tyrosine between the N- and C-terminal tyrosines. Mutation of the N-terminal tyrosine (Y218) results in destabilization of the association between Lyn and β [8]. β spans the plasma membrane four times so that its N- and C- terminal regions are both in contact with the cytoplasm [9]:



In the above diagram, numbering of amino acid residues is based on UniProt entry P13386 (uniprot.org).

(c) γ **chain.** When phosphorylated, the γ chain recruits the kinase Syk. The tandem SH2 domains of Syk bind the two ITAM tyrosine residues of a γ chain [10–13]:



In the above diagram, numbering of amino acid residues is based on UniProt entry P20411 (uniprot.org).

In the model (model.bngl, ESI) FcϵRI (`Rec`) is assigned the following molecule type definition:

```
Rec(a,b~0~P,g~0~P)
```
(2)

The component `a` represents the α chain of FcϵRI, the component `b` represents the β chain of FcϵRI, and the component `g` represents the γ chains of FcϵRI. The `b` and `g` component each have two possible states: `0` (unphosphorylated) and `P` (phosphorylated). The initial condition of both is `0` (Table S3). The two γ chains are modeled as a single unit. The α chain is mostly extracellular, whereas the β and γ chains have significant cytoplasmic regions. In the model, `Rec` is located in the plasma membrane compartment, `M` (Table S2). The total abundance of `Rec` is `Rec_tot` (Table S3). In the extended contact map, FcϵRI is represented by a simplified form of the following annotated hierarchical molecule type graph:



3. **Lyn (protein tyrosine kinase)**

Lyn is a Src-family kinase that is activated following FcεRI engagement [14]. Lyn is essential for phosphorylation of substrates in response to receptor stimulation [15]. Its unique N-terminal region associates weakly with the unphosphorylated β chain [14, 16–19]. Like other members of the Src family, Lyn contains an SH3 domain, an SH2 domain, and a protein tyrosine kinase (PTK) domain:



In the above diagram, numbering of amino acid residues is based on UniProt entry Q07014 (uniprot.org).

In the model (model.bngl, ESI) Lyn (`Lyn`) is assigned the following molecule type definition:

```
Lyn(U,SH2)
```
(3)

The component `U` represents the unique N-terminal region of Lyn, and the component `SH2` represents the SH2 domain of Lyn. Lyn is anchored to the inner leaflet of the plasma membrane. In the model,

`Lyn` is located in the plasma membrane compartment, `M` (Table S2). The total abundance of `Lyn` is `Lyn_tot` (Table S3).

In the extended contact map, Lyn is represented by a simplified form of the following annotated hierarchical molecule type graph:



4. **Syk (spleen tyrosine kinase)**

Syk is the prototypical member of the Syk-family of protein tyrosine kinases [20]. It plays an important role in FcϵRI signaling [21]. Syk contains two tandem SH2 domains, a PTK domain, and multiple sites of phosphorylation, including Y317, Y342 and Y346 in the linker region and Y519 and Y520 in the activation loop of the PTK domain. As documented in Phospho.ELM [22], Y317, Y342, and Y346 are primarily phosphorylated by Lyn [23, 24] and Y519 and Y520 are primarily transphosphorylated by Syk [23]. Phosphorylation of Y130, located between the two SH2 domains, may have a role in regulation of catalytic activity and dissociation from the receptor [25]. However, this site is not considered in this model. Thus, it is not shown in the extended contact map of Fig. 4 in the main text.



In the above diagram, numbering of amino acid residues is based on UniProt entry Q64725

(uniprot.org).

In the model (model.bngl, ESI), Syk (`Syk`) is assigned the following molecule type definition:

```
Syk(tSH2,l~0~P,a~0~P)
```
(4)

The component `tSH2` represents the tandem SH2 domains of Syk, the component `l` represents tyrosine residues in the linker region of Syk, and the component `a` represents tyrosine residues in the activation loop of the PTK domain in Syk. The components `l` and `a` each have two possible internal states: `0` (not phosphorylated) and `P` (phosphorylated). The initial condition of both is `0`. In the absence of interactions with other molecules, Syk is located in the cytoplasmic compartment, `C` (Table S2). The total abundance of `Syk` is `Syk_tot` (Table S3).

In the extended contact map (Fig. 4), Syk is represented by a simplified form of the following annotated hierarchical molecule type graph:



Notes on Interactions Included in Model

1. **The IgE dimer binds the α subunit of FcϵRI.** The α chain is the subunit of the receptor that binds IgE. The IgE dimer interacts with FcϵRI to mediate the formation of receptor dimers, which stimulate mast-cell degranulation [3]. The kinetics of receptor signaling stimulated by IgE dimer has been studied [5]. The interaction of IgE with FcϵRI is modeled using the following rules:

```
Rec(a) + Lig(Fc,Fc) <->
Rec(a!1).Lig(Fc!1,Fc) kp1, km1
```
(1a)

```
Rec(a) + Lig(Fc,Fc!1) <->
Rec(a!2).Lig(Fc!2,Fc!1) kp2,km2
```
(1b)

where the first rule characterizes initial binding of ligand and receptor, and the second rule characterizes receptor aggregation.

2. **The β subunit of FcϵRI binds the unique domain in Lyn.** Lyn preferentially interacts with the C-terminal region of the β chain [16,18,19]. This interaction does not require receptor stimulation, nor does it require the ITAM tyrosine residues [17]. The constitutive interaction of the β subunit and Lyn is modeled using the following rule:

```
Rec(b~0) + Lyn(U,SH2) <->
Rec(b~0!1).Lyn(U!1,SH2) kpL, kmL
```
(2)

3. **Lyn transphosphorylates Y218, Y225, and Y229 in the β subunit of FcϵRI.** The β chain of FcϵRI becomes phosphorylated upon receptor aggregation and dephosphorylated upon disaggregation [26]. When bound to a receptor, Lyn *trans*-phosphorylates the β and γ ITAMs in neighboring receptors [27]. Lyn is the primary kinase responsible for phosphorylation of these sites [15,27]. Constitutively-associated Lyn is capable of phosphorylating these residues [7]. Upon receptor aggregation and phosphorylation, the SH2 domain of Lyn is able to bind β; the result is stronger Lyn-receptor association and increased Lyn kinase activity [16,27]. Accordingly, phosphorylation of the β subunit by Lyn is modeled using two rules, one for each of the receptor-bound forms of Lyn:

```
Lig(Fc!1,Fc!2).Lyn(U!3,SH2).Rec(a!2,b~0!3).Rec(a!1,b~0) ->
Lig(Fc!1,Fc!2).Lyn(U!3,SH2).Rec(a!2,b~0!3).Rec(a!1,b~P) pLb
```
(3a)

```
Lig(Fc!1,Fc!2).Lyn(U,SH2!3).Rec(a!2,b~P!3).Rec(a!1,b~0) ->
Lig(Fc!1,Fc!2).Lyn(U,SH2!3).Rec(a!2,b~P!3).Rec(a!1,b~P) pLbs
```
(3b)

where the first rule characterizes phosphorylation by constitutive Lyn and the second rule characterizes phosphorylation by SH2-bound Lyn (see Interaction 5). Tyrosines in the β ITAM are lumped together as a single site.

4. **Lyn transphosphorylates Y65 and Y76 in the γ subunit of FcϵRI.** The γ chains of FcϵRI become phosphorylated upon receptor aggregation and dephosphorylated upon receptor disaggregation [26]. When bound to a receptor, Lyn may phosphorylate β and γ subunits [27]. Lyn-deficient mast cells show impaired tyrosine phosphorylation [15]. Phosphorylation of the γ subunit by Lyn is modeled using the following rules:

```
Lig(Fc!1,Fc!2).Lyn(U!3,SH2).Rec(a!2,b~0!3).Rec(a!1,g~0) ->
Lig(Fc!1,Fc!2).Lyn(U!3,SH2).Rec(a!2,b~0!3).Rec(a!1,g~P) pLg
```
(4a)

```
Lig(Fc!1,Fc!2).Lyn(U,SH2!3).Rec(a!2,b~P!3).Rec(a!1,g~0) ->
Lig(Fc!1,Fc!2).Lyn(U,SH2!3).Rec(a!2,b~P!3).Rec(a!1,g~P) pLgs
```
(4b)

where the first rule characterizes phosphorylation by constitutive Lyn and the second rule characterizes phosphorylation by SH2-bound Lyn (see Interaction 5). Tyrosines in the ITAMs of the two γ chains are lumped together as a single site.

5. **The phosphorylated β subunit binds the SH2 domain in Lyn.** Association of Lyn with the β subunit increases following receptor aggregation, paralleling the aggregation-induced increase in receptor phosphorylation [16]. Recruitment of Lyn via its SH2 domain results in a much stronger association than constitutive binding of Lyn to the unphosphorylated receptor [11, 16, 27]. There is evidence that the N-terminal ITAM tyrosine (Y218) is the most critical residue for this interaction [8]. The SH2-based interaction of the β subunit and Lyn is modeled using the following rule:

```
Rec(b~P) + Lyn(U,SH2) <->
Rec(b~P!1).Lyn(U,SH2!1) kpLs, kmLs
```
(5)

6. **The phosphorylated γ subunit binds the SH2 domain in Syk.** Syk is recruited to the receptor primarily by its tandem SH2 domains binding the doubly-phosphorylated γ ITAM [10–13]. The affinity of Syk for the doubly-phosphorylated γ ITAM is at least 1,000 times higher than its affinity for singly-phosphorylated γ [10], and at least 20 times higher than its affinity for doubly-phosphorylated β [13]. The interaction of the γ subunit and Syk is modeled using the following rule:

```
Rec(g~P) + Syk(tSH2) <->
Rec(g~P!1).Syk(tSH2!1) kpS, kmS
```
(6)

The γ ITAM tyrosines are lumped together as a single site, as are the two SH2 domains of Syk.

7. **Lyn transphosphorylates Y317, Y342, and Y346 in the linker region in Syk.** Lyn is primarily responsible for phosphorylating residues in the linker region, which includes a negative regulatory site (Y317) [23, 24]. Phosphorylation of Syk by Lyn is modeled using the following rules:

```
Lig(Fc!1,Fc!2).Lyn(U!3,SH2).Rec(a!2,b~0!3).
Rec(a!1,g~P!4).Syk(tSH2!4,l~0) ->
Lig(Fc!1,Fc!2).Lyn(U!3,SH2).Rec(a!2,b~0!3).
Rec(a!1,g~P!4).Syk(tSH2!4,l~P) pLS
```
(7a)

```
Lig(Fc!1,Fc!2).Lyn(U,SH2!3).Rec(a!2,b~P!3).
Rec(a!1,g~P!4).Syk(tSH2!4,l~0) ->
Lig(Fc!1,Fc!2).Lyn(U,SH2!3).Rec(a!2,b~P!3).
Rec(a!1,g~P!4).Syk(tSH2!4,l~P) pLSs
```
(7b)

where the first rule characterizes phosphorylation by constitutive Lyn and the second rule characterizes phosphorylation by SH2-bound Lyn. All tyrosine residues in the linker region are lumped together as a single site.

8. **Syk transphosphorylates Y519 and Y520 in the activation loop in Syk.** Full activation of Syk, needed for degranulation of RBL cells, requires transphosphorylation at the two activation loop tyrosines, Y519 and Y520 [28–30]. Phosphorylation of Syk by Syk is modeled using the following rules:

```
Lig(Fc!1,Fc!2).Syk(tSH2!3,a~0).Rec(a!2,g~P!3).
Rec(a!1,g~P!4).Syk(tSH2!4,a~0) ->
Lig(Fc!1,Fc!2).Syk(tSH2!3,a~0).Rec(a!2,g~P!3).
Rec(a!1,g~P!4).Syk(tSH2!4,a~P) pSS
```
(8a)

```
Lig(Fc!1,Fc!2).Syk(tSH2!3,a~P).Rec(a!2,g~P!3).
Rec(a!1,g~P!4).Syk(tSH2!4,a~0) ->
Lig(Fc!1,Fc!2).Syk(tSH2!3,a~P).Rec(a!2,g~P!3).
Rec(a!1,g~P!4).Syk(tSH2!4,a~P) pSSs
```
(8b)

where the first rule characterizes phosphorylation by Syk that is not phosphorylated on its activation loop, and the second rule characterizes phosphorylation by Syk that is phosphorylated on its activation loop. The two residues are lumped together as a single site.

9. **Dephosphorylation by unspecified phosphatases.** Phosphorylated tyrosine residues are desphosphorylated by a pool of phosphatases. Dephosphorylation of a site is largely blocked when an SH2 domain binds the site [31]. Accordingly, only unbound phosphorylated residues are treated as substrates for phosphatases. Dephosphorylation is modeled using the following rules:

```
Rec(b~P)-> Rec(b~0) dm
```
(9a)

```
Rec(g~P)-> Rec(g~0) dm
```
(9b)

```
Syk(tSH2!1,l~P)-> Syk(tSH2!1,l~0) dm
```
(9c)

```
Syk(tSH2!1,a~P)-> Syk(tSH2!1,a~0) dm
```
(9d)

```
Syk(tSH2,l~P)-> Syk(tSH2,l~0) dc
```
(9e)

```
Syk(tSH2,a~P)-> Syk(tSH2,a~0) dc
```
(9f)

where the first and second rules characterize dephosphorylation of the β and γ chains, respectively. The third and fourth rules characterize dephosphorylation of the linker region and activation loop of membrane-recruited Syk, and the fifth and sixth rules characterize dephosphorylation of cytosolic Syk.

Table S2

Compartments					
Compartment	**Dimension**	**Enclosing compartment**	**Parameter**	**Value**	**Reference**
E	3	n/a	v_E	$1*10^{-6}$ml	[2]
M	2	E	sa_M	600 μm^2	[2]
C	3	M	v_C	$1.4*10^{-9}$ ml	[2]

An effective membrane volume, ev_M, is obtained by multiplying surface area, sa_M, by an effective width, eff_width $=1*10^{-2}\mu$m. Note that v_E is the inverse of the cell density at which experiments were performed (10^6 cells/ml).

Table S3

Seed Species				
Molecule	**Initial condition**	**Amount**	**Value**	**Reference**
Lig(Fc,Fc)	Lig(Fc,Fc)@E	Lig_tot	6.0 * 10^3/cell	
Rec(a,b∼0∼P,g∼0∼P)	Rec(a,b∼0,g∼0)@M	Rec_tot	4.0 * 10^5/cell	[2]
Lyn(U,SH2)	Lyn(U,SH2)@M	Lyn_tot	2.8 * 10^4/cell	Available Lyn∼0.07 x RT [32]
Syk(tSH2,l∼0∼P,a∼0∼P)	Syk(tSH2,l∼0,a∼0)@C	Syk_tot	4.0 * 10^5/cell	[2]

Table S4

Kinetic parameters		
1a: Ligand-receptor binding		
kp1	1.3 * 10^{-10} molecules^{-1}s^{-1}*v_E	Estimated binding parameters for covalently cross-linked IgE dimer [32]
km1	0	10^{-5} s^{-1} $\sim$ 0 over reported time scales
1b: Receptor aggregation		
kp2	2.5* 10^{-4} molecules^{-1}s^{-1}*ev_M	$k_{+2}R_T$=100 s^{-1} [32]
km2	0	10^{-5} s^{-1} $\sim$ 0 over reported time scales
2: Binding of Lyn through its unique domain to beta subunit		
kpL	5 * 10^{-5} molecules^{-1}s^{-1} $*$ ev_M	Based on estimated equilibrium constants in [32]
kmL	20 s^{-1}	Based on estimated equilibrium constants in [32]
3a: Phosphorylation of β subunit by constitutive Lyn		
pLb	30 s^{-1}	Consistent with extensive receptor phosphorylation
3b: Phosphorylation of β subunit by SH2-bound Lyn		
pLbs	100 s^{-1}	Moderate increase in Lyn kinase activity upon SH2 domain binding [27]
4a: Phosphorylation of γ subunit by constitutive Lyn		
pLg	1 s^{-1}	Double phosphorylation of γ ITAM tyrosine, required to bind Syk, slower than single phosphorylation [33–35]
4b: Phosphorylation of γ subunit by SH2-bound Lyn		
pLgs	3 s^{-1}	Moderate increase in Lyn kinase activity upon SH2 domain binding [27]
5: Binding of Lyn through its SH2 domain to phosphorylated β		
kpLs	5 * 10^{-5} molecules$^{-1}s^{-1}$*ev_M	Based on estimated equilibrium constants in [32]
kmLs	0.12 s^{-1}	Fit to observed rate of β ITAM dephosphorylation from [36]
6: Binding of Syk to phosphorylated γ		
kpS	6 * 10^{-5} molecules^{-1} s^{-1}*v_C	Based on measured equilibrium constant at 25°C for binding of Syk tandem SH2 domains [13]
kmS	0.13 s-1	Fit to observed rate of γ ITAM dephosphorylation in [36]
7a: Phosphorylation of Syk by constitutive Lyn		
pLS	30 s^{-1}	Consistent with extensive receptor phosphorylation
7b: Phosphorylation of Syk by SH2-bound Lyn		
pLSs	100 s^{-1}	Moderate increase in Lyn kinase activity upon SH2 domain binding [27]
8a: Phosphorylation of Syk by Syk		
pSS	100 s^{-1}	Assume same kinase activity as recruited Lyn
8b: Phosphorylation of Syk by activated Syk		
pSSs	200 s^{-1}	Moderate increase in Syk activity upon phosphorylation of activation loop [29]
9: Dephosphorylation		
dm	20 s^{-1}	Fit to rate of ITAM dephosphorylation in [36]
dc	20 s^{-1}	Fit to rate of ITAM dephosphorylation in [36]

References

1. Goldstein B, Faeder JR, Hlavacek WS, Blinov ML, Redondo A, Wofsy C: **Modeling the early signaling events mediated by FcεRI**. *Mol Immunol* 2002, **38**:1213–1219.
2. Faeder JR, Hlavacek WS, Reischl I, Blinov ML, Metzger H, Redondo A, Wofsy C, Goldstein B: **Investigation of early events in FcεRI-mediated signaling using a detailed mathematical model**. *J Immunol* 2003, **170**:3769–3781.
3. Segal DM, Taurog JD, Metzger H: **Dimeric immunoglobulin E serves as a unit signal for mast cell degranulation**. *Proc Natl Acad Sci USA* 1977, **74**:2993–2997.
4. Kent UM, Mao SY, Wofsy C, Goldstein B, Ross S, Metzger H: **Dynamics of signal transduction after aggregation of cell-surface receptors: studies on the type I receptor for IgE**. *Proc Natl Acad Sci USA* 1994, **91**:3087–3091.
5. Wofsy C, Kent UM, Mao SY, Metzger H, Goldstein B: **Kinetics of tyrosine phosphorylation when IgE dimers bind to FCε receptors on rat basophilic leukemia cells**. *J Biol Chem* 1995, **270**:20264–20272.
6. Kraft S, Kinet JP: **New developments in FcεRI regulation, function, and inhibition**. *Nat Rev Immunol* 2007, **7**:365–378.
7. Lin S, Cicala C, Scharenberg AM, Kinet JP: **The FcεRIβ subunit functions as an amplifier of FcεRIγ-mediated cell activation signals**. *Cell* 1996, **85**:985–995.
8. On M, Billingsley JM, Jouvin MH, Kinet JP: **Molecular dissection of the FcRβ signaling amplifier**. *J Biol Chem* 2004, **279**:45782–45790.
9. Küster H, Zhang L, Brini AT, MacGlashan DW, Kinet JP: **The gene and cDNA for the human high affinity immunoglobulin E receptor β chain and expression of the complete human receptor**. *J Biol Chem* 1992, **267**:12782–12787.
10. Chen T, Repetto B, Chizzonite R, Pullar C, Burghardt C, Dharm E, Zhao Z, Carroll R, Nunes P, Basu M, Danho W, Visnick M, Kochan J, Waugh D, Gilfillan AM: **Interaction of phosphorylated FcεRIγ immunoglobulin receptor tyrosine activation motif-based peptides with dual and single SH2 domains of p72^{syk}. Assessment of binding parameters and real time binding kinetics**. *J Biol Chem* 1996, **271**:25308–25315.
11. Kihara H, Siraganian RP: **Src homology 2 domains of Syk and Lyn bind to tyrosine-phosphorylated subunits of the high affinity IgE receptor**. *J Biol Chem* 1994, **269**:22427–22432.
12. Shiue L, Green J, Green OM, Karas JL, Morgenstern JP, Ram MK, Taylor MK, Zoller MJ, Zydowsky LD, Bolen JB, Brugge JS: **Interaction of p72^{syk} with the γ and β subunits of the high-affinity receptor for immunoglobulin E, FcεRI**. *Mol Cell Biol* 1995, **15**:272–281.
13. Ottinger EA, Botfield MC, Shoelson SE: **Tandem SH2 domains confer high specificity in tyrosine kinase signaling**. *J Biol Chem* 1998, **273**:729–735.
14. Eiseman E, Bolen JB: **Engagement of the high-affinity IgE receptor activates *src* protein-related tyrosine kinases**. *Nature* 1992, **355**:78–80.
15. Nishizumi H, Yamamoto T: **Impaired tyrosine phosphorylation and Ca^{2+} mobilization, but not degranulation, in Lyn-deficient bone marrow-derived mast cells**. *J Immunol* 1997, **158**:2350–2355.
16. Yamashita T, Mao SY, Metzger H: **Aggregation of the high-affinity IgE receptor and enhanced activity of p53/56^{lyn} protein-tyrosine kinase**. *Proc Natl Acad Sci USA* 1994, **91**:11251–11255.
17. Jouvin MH, Adamczewski M, Numerof R, Letourneur O, Valle A, Kinet JP: **Differential control of the tyrosine kinases Lyn and Syk by the two signaling chains of the high affinity immunoglobulin E receptor**. *J Biol Chem* 1994, **269**:5918–5925.
18. Vonakis BM, Chen H, Haleem-Smith H, Metzger H: **The unique domain as the site on Lyn kinase for its constitutive association with the high affinity receptor for IgE**. *J Biol Chem* 1997, **272**:24072–24080.
19. Vonakis BM, Haleem-Smith H, Benjamin P, Metzger H: **Interaction between the unphosphorylated receptor with high affinity for IgE and Lyn kinase**. *J Biol Chem* 2001, **276**:1041–1050.
20. Bradshaw JM: **The Src, Syk, and Tec family kinases: distinct types of molecular switches**. *Cell Signal* 1992, **22**:1175–1184.

21. Gilfillan AM, Rivera J: **The tyrosine kinase network regulating mast cell activation**. *Immunol Rev* 2009, **228**:149–169.

22. **Phospho.ELM** [http://phospho.elm.eu.org].

23. Keshvara LM, Isaacson CC, Yankee TM, Sarac R, Harrison ML, Geahlen RL: **Syk- and Lyn-dependent phosphorylation of Syk on multiple tyrosines following B cell activation includes a site that negatively regulates signaling**. *J Immunol* 1998, **161**:5276–5283.

24. Hong JJ, Yankee TM, Harrison ML, Geahlen RL: **Regulation of signaling in B cells through the phosphorylation of Syk on linker region tyrosines. A mechanism for negative signaling by the Lyn tyrosine kinase**. *J Biol Chem* 2002, **277**:31703–31714.

25. Keshvara LM, Isaacson C, Harrison ML, Geahlen RL: **Syk activation and dissociation from the B-cell antigen receptor is mediated by phosphorylation of tyrosine 130**. *J Biol Chem* 1997, **272**:10377–10381.

26. Paolini R, Jouvin MH, Kinet JP: **Phosphorylation and dephosphorylation of the high-affinity receptor for immunoglobulin E immediately after receptor engagement and disengagement**. *Nature* 1991, **353**:855–858.

27. Pribluda VS, Pribluda C, Metzger H: **Transphosphorylation as the mechanism by which the high-affinity receptor for IgE is phosphorylated upon aggregation**. *Proc Natl Acad Sci USA* 1994, **91**:11246–11250.

28. Zhang J, Kimura T, Siraganian RP: **Mutations in the activation loop tyrosines of protein tyrosine kinase Syk abrogate intracellular signaling but not kinase activity**. *J Immunol* 1998, **161**:4366–4374.

29. Zhang J, Billingsley ML, Kincaid RL, Siraganian RP: **Phosphorylation of Syk activation loop tyrosines is essential for Syk function. An *in vivo* study using a specific anti-Syk activation loop phosphotyrosine antibody**. *J Biol Chem* 2000, **275**:35442–35447.

30. Siraganian RP, Zhang J, Suzuki K, Sada K: **Protein tyrosine kinase Syk in mast cell signaling**. *Mol Immunol* 2002, **38**:1229–1233.

31. Rotin D, Margolis B, Mohammadi M, Daly RJ, Daum N G Li, Fischer EH, Burgess WH, Ullrich A, Schlessinger J: **SH2 domains prevent tyrosine dephosphorylation of the EGF receptor: identification of Tyr992 as the high-affinity binding site for SH2 domains of phospholipase Cγ**. *EBMO J* 1992, **11**:559–567.

32. Wofsy C, Torigoe C, Kent UM, Metzger H, Goldstein B: **Exploiting the difference between intrinsic and extrinsic kinases: implications for regulation of signaling by immunoreceptors**. *J Immunol* 1997, **159**:2984–2992.

33. Pao LI, Famiglietti SJ, Cambier JC: **Asymmetrical phosphorylation and function of immunoreceptor tyrosine-based activation motif tyrosines in B cell antigen receptor signal transduction**. *J Immunol* 1998, **160**:3305–3314.

34. Gaul BS, Harrison ML, Geahlen RL, Burton RA, Post CB: **Substrate recognition by the Lyn protein-tyrosine kinase. NMR structure of the immunoreceptor tyrosine-based activation motif signaling region of the B cell antigen receptor**. *J Biol Chem* 2000, **275**:16174–16182.

35. Pribluda VS, Pribluda C, Metzger H: **Biochemical evidence that the phosphorylated tyrosines, serines, and threonines on the aggregated high affinity receptor for IgE are in the immunoreceptor tyrosine-based activation motifs**. *J Biol Chem* 1997, **272**:11185–11192.

36. Mao SY, Metzger H: **Characterization of protein-tyrosine phosphatases that dephosphorylate the high affinity IgE receptor**. *J Biol Chem* 1997, **30**:14067–14073.